X04BMFP

NAG Parallel Library Routine Document

Note: before using this routine, please read the Users' Note for your implementation to check for implementation-dependent details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

1 Description

X04BMFP outputs a set of integer matrices distributed on a two-dimensional logical processor grid. It writes to output units connected to the root processor.

Each matrix on a logical processor is nominally denoted by A. Each matrix A and the row dimension m and the column dimension n are local to each logical processor. In general, each m and each n may be different on different logical processors in the Library Grid. This routine does not assume that the local matrices A are part of any particular global data structure. It prints the local matrices in the row-major ordering of the grid.

In general, each matrix A can be printed on a different output unit NOUT defined on the logical processor which owns the matrix A but in practice NOUT will generally have the same value on all logical processors and the matrices will then be written to a single file.

Optionally a standard title which identifies the logical processor can be printed followed by a blank line before a matrix is printed. As an example, the title

```
Array from logical processor {5,3}
```

can be printed before the matrix on logical processor {5,3}. The matrices are output with a **maximum record length of 80**. Options are also available to print column numbers of the matrices.

If the root processor is not present on the current Library Grid then the logical processor {0,0} of the current two-dimensional grid takes over the role of the root processor.

2 Specification

```
 SUBROUTINE X04BMFP(ICNTXT, NOUT, M, N, A, LDA, FORMAT, TITOP,
1                   CNUMOP, ICOFF, W, LDW, IFAIL)
 INTEGER            ICNTXT, NOUT, M, N, A(LDA,*), LDA, ICOFF,
1                   W(LDW,*), LDW, IFAIL
 CHARACTER*1        TITOP, CNUMOP
 CHARACTER*(*)      FORMAT
```

3 Data Distribution

3.1 Definitions

The following definitions are used in describing the data distribution within this document:

m	–	the number of rows of a matrix on a logical processor.
n	–	the number of columns of a matrix on a logical processor.
$m_{\max}$	–	the maximum value of m taken over all logical processors (which have a matrix to be printed) except the root processor.
$n_{\max}$	–	the maximum value of n taken over all logical processors (which have a matrix to be printed) except the root processor.

3.2 Global and Local Arguments

The following global **input** arguments must have the same value on entry to the routine on each processor and the global **output** arguments will have the same value on exit from the routine on each processor:

Global input arguments: IFAIL

Global output arguments: IFAIL

The remaining arguments are local.

Note: the workspace argument W and the input arguments FORMAT, TITOP, and CNUMOP are referenced only on the root (or {0,0}) processor.

3.3 Distribution Strategy

All integer matrices are defined locally on each logical processor. The values of m and n (which define the dimensions of the matrices) do not have to be identical on every logical processor.

4 Arguments

1: ICNTXT — INTEGER *Local Input*

On entry: the Library context, usually returned by a call to the Library Grid initialisation routine Z01AAFP.

Note: the value of ICNTXT **must not** be changed.

2: NOUT — INTEGER *Local Input*

On entry: unit number for output. If NOUT is non-positive, no output of this matrix (including the standard title which specifies the processor) is done.

3: M — INTEGER *Local Input*

On entry: m, the number of rows of the local matrix. If m is non-positive, no output of this matrix (including the standard title which specifies the processor) is done.

4: N — INTEGER *Local Input*

On entry: n, the number of columns of the local matrix. If n is non-positive, no output of this matrix (including the standard title which specifies the processor) is done.

5: A(LDA,$*$) — INTEGER array *Local Input*

Note: the size of the second dimension of the array A must be at least max(1,N).

On entry: the local matrix A.

6: LDA — INTEGER *Local Input*

On entry: the size of the first dimension of the array A as declared in the (sub)program from which X04BMFP is called.

Constraint: LDA $\geq$ M if M $>$ 0, N $>$ 0 and NOUT $>$ 0; otherwise LDA $\geq$ 1.

7: FORMAT — CHARACTER*(*) *Local Input*

On entry: a valid Fortran format code on the root processor. This may be any format code allowed on the root processor, whether it is standard Fortran or not. FORMAT is used to output elements of the matrices. It may or may not be enclosed in brackets. Examples of valid values for FORMAT are `'(I6)', 'I4,2X'`.

FORMAT is not referenced on other logical processors.

Constraint: The character length of FORMAT must not be greater than 80.

8: TITOP — CHARACTER*1 *Local Input*

On entry: if TITOP = 'Y', then the standard title which identifies the logical processor is printed before the corresponding matrix is printed. Otherwise the standard title is not printed.

TITOP is referenced only by the root processor.

9: CNUMOP — CHARACTER*1 *Local Input*

On entry: indicates the type of labelling to be applied to columns of the matrices.

If CNUMOP = 'L' then the numbering of columns is specific to the matrix being printed. The columns are numbered starting at ICOFF + 1 and ending at ICOFF + n for each matrix.

If CNUMOP = 'G' then the columns are numbered consecutively in the order they are printed. The columns are numbered starting at 1 and ending at n_{total} where n_{total} is the sum of all (positive) values of n.

If CNUMOP $\neq$ 'L' or 'G' then the columns are not numbered.

CNUMOP is referenced only by the root processor.

10: ICOFF — INTEGER *Local Input*

On entry: offset value for column numbering if CNUMOP = 'L'.

11: W(LDW,$*$) — INTEGER array *Local Workspace*

Note: the size of the second dimension of the array W must be at least max(1,$n_{\max}$). This array is referenced only on the root processor.

12: LDW — INTEGER *Local Input*

On entry: the size of the first dimension of the array W as declared in the (sub)program from which X04BMFP is called.

Constraint: on the root processor LDW must be at least max(1,$m_{\max}$).

13: IFAIL — INTEGER *Global Input/Global Output*

The NAG Parallel Library provides a mechanism, via the routine Z02EAFP, to reduce the amount of parameter validation performed by this routine. For a full description refer to the Z02 Chapter Introduction.

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this argument (described in the Essential Introduction) the recommended values are:

IFAIL = 0, if multigridding is **not** employed;
IFAIL = −1, if multigridding is employed.

On exit: IFAIL = 0 (or −9999 if reduced error checking is enabled) unless the routine detects an error (see Section 5).

5 Errors and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output from the root processor (or processor {0,0} when the root processor is not available) on the current error message unit (as defined by X04AAF).

5.1 Full Error Checking Mode Only

IFAIL = −2000

The routine has been called with an invalid value of ICNTXT on one or more processors.

IFAIL = −1000

The logical processor grid and library mechanism (Library Grid) have not been correctly defined, see Z01AAFP.

IFAIL = $-i$

The ith argument had an invalid value. This error occurred either because a global argument did not have the same value on all logical processors, or because its value on one or more processors was incorrect. An explanatory message distinguishes between these two cases.

5.2 Any Error Checking Mode

IFAIL = 1

On entry, variable FORMAT is more than 80 characters long.

IFAIL = 2

The code supplied in FORMAT cannot be used to output a number. FORMAT probably has too wide a field width or contains an illegal edit descriptor.

6 Further Comments

6.1 Algorithmic Detail

None.

6.2 Parallelism Detail

Each matrix is brought to the root processor and is then printed. This is performed sequentially using the row-major ordering of the grid.

7 References

None.

8 Example

A different matrix is generated on each processor in the grid and then they are printed on the root processor.

8.1 Example Text

```
*     X04BMFP Example Program Text
*     NAG Parallel Library Release 3. NAG Copyright 1999.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MG, NG
      PARAMETER        (MG=2,NG=2)
      INTEGER          LDA, TDA
      PARAMETER        (LDA=5,TDA=5)
      CHARACTER*20     FORMT
      PARAMETER        (FORMT='I5')
*     .. Local Scalars ..
      INTEGER          I, ICNTXT, ICOFF, IFAIL, J, M, MP, MYCOL, MYROW,
     +                 N, NP, NPCOL, NPROW
      LOGICAL          ROOT
      CHARACTER        CNUMOP, TITOP
*     .. Local Arrays ..
      INTEGER          A(LDA,TDA), W(LDA,TDA)
*     .. External Functions ..
      LOGICAL          Z01ACFP
      EXTERNAL         Z01ACFP
*     .. External Subroutines ..
      EXTERNAL         X04BMFP, Z01AAFP, Z01ABFP, Z01ZAFP
*     .. Executable Statements ..
      ROOT = Z01ACFP()
      IF (ROOT) THEN
```

```
         WRITE (NOUT,*) 'X04BMFP Example program Results'
         WRITE (NOUT,*)
      END IF
*
*     Define the 2D processor grid
*
      MP = MG
      NP = NG
      IFAIL = 0
*
      CALL Z01AAFP(ICNTXT,MP,NP,IFAIL)
*
      IFAIL = 0
*
*     Generate a set of matrices A on each processor
*
      CALL Z01ZAFP(ICNTXT,NPROW,NPCOL,MYROW,MYCOL)
      M = 2
      N = 3
      DO 20 J = 1, N
         DO 20 I = 1, M
            A(I,J) = I*MYROW + J*MYCOL
   20 CONTINUE
*
      IF (ROOT) THEN
         WRITE (NOUT,*) 'The matrices'
         WRITE (NOUT,*)
         TITOP = 'Y'
         CNUMOP = 'N'
      END IF
      ICOFF = 0
      IFAIL = 0
*
      CALL X04BMFP(ICNTXT,NOUT,M,N,A,LDA,FORMT,TITOP,CNUMOP,ICOFF,W,LDA,
     +             IFAIL)
*
*     Undefine the 2D processor grid
*
      IFAIL = 0
*
      CALL Z01ABFP(ICNTXT,'N',IFAIL)
*
      STOP
      END
```

8.2 Example Data

None.

8.3 Example Results

```
X04BMFP Example program Results

The matrices

  Array from logical processor    0,   0

     0    0    0
     0    0    0

  Array from logical processor    0,   1

     1    2    3
     1    2    3

  Array from logical processor    1,   0

     1    1    1
     2    2    2

  Array from logical processor    1,   1

     2    3    4
     3    4    5


X04BMFP Example program Results
```